

บทที่ 5

การออกแบบโปรแกรมหน้าจอการแสดงผลและการพัฒนาโปรแกรม

จากการวิเคราะห์ฐานข้อมูลในบทที่ผ่านมาโดยใช้แผนภาพบริบท และแผนภาพการไหลของข้อมูล รวมถึงแผนภาพความสัมพันธ์ของเอนทิตี การออกแบบฐานข้อมูล เพื่อให้เห็นความสัมพันธ์ของข้อมูลในกระบวนการต่างๆ และให้สอดคล้องกับความต้องการของผู้ใช้งาน ในบทนี้จึงได้กล่าวถึงการออกแบบโครงสร้างโปรแกรม และหน้าจอการแสดงผลในการจัดการสินค้าคงคลังของร้าน ส.อะไหล่ จังหวัดเพชรบูรณ์ โดยเก็บข้อมูลจากผู้ใช้งานถึงความต้องการในโครงสร้างโปรแกรม และหน้าจอการแสดงผล และออกแบบให้มีความสัมพันธ์กับออกแบบฐานข้อมูลในบทที่ผ่านมา เพื่อให้ผู้ใช้งานมีความสะดวกในการใช้งาน โดยแบ่งเป็น

- 1) โครงสร้างโปรแกรม
- 2) การออกแบบหน้าจอและการแสดงผล
- 3) การพัฒนาโปรแกรม

5.1 โครงสร้างโปรแกรม

การออกแบบโครงสร้างโปรแกรมได้แบ่งตามสิทธิของผู้ใช้งาน โดยแบ่งตามหน้าที่ที่รับผิดชอบ ได้แก่ พนักงาน ผู้บริหาร และผู้ดูแลระบบ โดยผู้ใช้งานจะต้องล็อกอินเข้าสู่ระบบทุกครั้งก่อนการใช้งาน เมื่อล็อกอินเข้าสู่ระบบแล้ว ผู้ใช้งานสามารถแก้ไขข้อมูลส่วนตัว และรหัสผ่านของตนเองได้ ส่วนการจัดการข้อมูลอื่นๆ สิทธิในการจัดการข้อมูลจะแตกต่างกันไปตามการกำหนดสิทธิของผู้ดูแลระบบ หรือผู้บริหารที่มีสิทธิสูงกว่า เช่น ผู้ดูแลระบบ สามารถจัดการข้อมูลเบื้องต้น ได้แก่ ข้อมูลประเภทสินค้า จัดการข้อมูลผู้จำหน่าย ข้อมูลสินค้า ข้อมูลพนักงาน สิทธิการใช้งานของพนักงาน รวมไปถึงการสำรองข้อมูล พนักงานสามารถจัดการข้อมูลสินค้า การสั่งซื้อ การตรวจรับสินค้าจากผู้จำหน่าย รายงานที่เกี่ยวกับการสั่งซื้อสินค้า สามารถจัดการข้อมูลจุดสั่งซื้อสินค้า ตรวจสอบสินค้าที่ถึงจุดสั่งซื้อ รายงานที่เกี่ยวกับคลังสินค้า จัดการข้อมูลลูกค้า การขายสินค้า ให้กับลูกค้า การเปลี่ยนสินค้าในกรณีที่สินค้าชำรุดจากทางร้าน หรือต้องการเปลี่ยนสินค้าใหม่โดยจะต้องอยู่ในช่วงเวลาที่สามารถเปลี่ยนสินค้าได้ และรายงานที่เกี่ยวกับการขายสินค้า ผู้บริหารสามารถดูรายงานทั้งหมดที่เกี่ยวข้องกับการสั่งซื้อ คลังสินค้า และการขายสินค้า

5.2 การออกแบบหน้าจอการแสดงผล

การออกแบบหน้าจอการแสดงผล ได้มีการสอบถามถึงความต้องการของผู้ใช้งาน รวมถึงการทำงานประจำวัน โดยมีจุดประสงค์เพื่อให้การออกแบบหน้าจอแสดงผลเหมาะสมต่อการใช้งานของผู้ใช้งาน และให้สอดคล้องกับโครงสร้าง โปรแกรมแสดงการออกแบบหน้าจอได้ดังนี้

1) หน้าจอเข้าสู่ระบบ

ส่วนที่ 1

ส่วนที่ 2

รหัสผู้ใช้:

รหัสผ่าน:

ร้านค้า:

Login Cancel

รูป 5.1 หน้าจอเข้าสู่ระบบ

จากรูป 5.1 สามารถอธิบายได้ ดังนี้

ส่วนที่ 1 แสดงส่วนของชื่อร้านค้า

ส่วนที่ 2 แสดงส่วนของการเข้าสู่ระบบจัดการสินค้าคงคลัง ร้านสะดวกซื้อ โดยมิสิทธิการเข้าสู่ระบบของพนักงานแต่ละบุคคลในการจัดการข้อมูลแตกต่างกันไปตามหน้าที่ ผู้ดูแลระบบ พนักงาน และผู้บริหาร ในทุกครั้งที่เข้าใช้งานต้องป้อนชื่อผู้ใช้ และรหัสผ่าน

2) หน้าจอระบบ

ส่วนที่ 1	ส่วนที่ 2
ส่วนที่ 3	ส่วนที่ 4
	ส่วนที่ 5

รูป 5.2 หน้าจอระบบ

จากรูป 5.2 สามารถอธิบายได้ ดังนี้

ส่วนที่ 1 แสดงรายละเอียดผู้ใช้ระบบ

ส่วนที่ 2 แสดงรายชื่อร้านค้า ชื่อหน้าจอที่ใช้ และสถานะการทำรายการ

ส่วนที่ 3 แสดงรายการที่ผู้ใช้ระบบสามารถทำรายการได้

ส่วนที่ 4 แสดงเมนูต่างๆ ในการทำรายการเกี่ยวกับหน้าจอ

ส่วนที่ 5 แสดงหน้าจอระบบต่างๆ ได้แก่

- หน้าจอจัดการหน้าจอระบบ
- หน้าจอจัดการผู้ใช้ระบบ
- หน้าจอจัดการหน้าจอระบบของผู้ใช้ระบบ
- หน้าจอใบทำรายการสั่งซื้อสินค้า
- หน้าจอใบทำรายการซื้อสินค้า
- หน้าจอใบทำรายการรับคืนสินค้า
- หน้าจอใบทำรายการขายสินค้า
- หน้าจอใบทำรายการส่งคืนสินค้า
- หน้าจอยกเลิกใบทำรายการ
- หน้าจอรายละเอียดสินค้า
- หน้าจอรายละเอียดลูกค้า

- หน้าจอประเภทลูกค้า
- หน้าจอประเภทการชำระ
- หน้าจอส่วนลดสินค้า
- หน้าจอสินค้าแถม
- หน้าจอประเภทสินค้า
- หน้าจอหน่วยนับ
- หน้าจอประเภทรายการ
- หน้าจอสินค้าย่อย
- หน้าจอประมวลผลจุดสั่งซื้อและยอดสั่งซื้อ
- หน้าจอชำระหนี้ลูกค้า
- หน้าจอออกรายงาน

3) หน้าจอค้นหาแบบลัด




รูป 5.3 หน้าจอค้นหาแบบลัด

จากรูป 5.3 สามารถอธิบายได้ ดังนี้

ส่วนที่ 1 แสดงรายการที่ค้นหา

ส่วนที่ 2 แสดงปุ่มการทำรายการ

4) หน้าจอค้นหา



รูป 5.4 หน้าจอค้นหา

จากรูป 5.4 สามารถอธิบายได้ ดังนี้

ส่วนที่ 1 แสดงรายละเอียดข้อมูลที่ต้องการค้นหา

ส่วนที่ 2 แสดงรายการที่ค้นหา

ส่วนที่ 3 แสดงปุ่มทำรายการ

5.3 การพัฒนาโปรแกรม

5.3.1 การพัฒนาโปรแกรมโดยโปรแกรม Sybase Power Builder 12.0 เป็นเครื่องมือในการพัฒนาโปรแกรมซึ่งพัฒนาโดยบริษัท Sybase โดยในการพัฒนาโปรแกรมด้วยเครื่องมือนี้จะประกอบไปด้วย Object ที่สำคัญ และเกี่ยวข้องกับระบบสารสนเทศเพื่อการจัดการร้าน ส.อะไหล่ จังหวัดเพชรบูรณ์ ที่สำคัญดังนี้

ตาราง 5.1 ตารางแสดง Object ที่สำคัญของ โปรแกรม PowerBuilder

ออบเจกต์	คำอธิบาย
workspace	เป็น Object ที่ใช้ในการกำหนดพื้นที่ที่ใช้ในการทำการพัฒนาระบบ
target	เป็นพื้นที่ที่ใช้ในการสร้าง Application หรือระบบ โดยใน 1 workspace สามารถเก็บ application ได้มากกว่า 1 application

ตาราง 5.1 ตารางแสดง Object ที่สำคัญของโปรแกรม PowerBuilder (ต่อ)

ออบเจกต์	คำอธิบาย
library	เป็นที่เก็บ PowerBuilder Object ต่างๆของโปรแกรม PowerBuilder ไม่ว่าจะเป็น window, datawindow, menu, function เป็นต้น
application	เป็นหัวใจหลักหรือแกนกลางในการพัฒนาโปรแกรมโดยโปรแกรม PowerBuilder ซึ่งใน application จะเป็น Object ที่ใช้ในการควบคุมการเริ่มต้นการทำงานของโปรแกรมที่ต้องการพัฒนา
window	เป็น Object ที่ใช้ในการติดต่อกับผู้ใช้งาน
datawindow	เป็น Object ที่ใช้ในการเชื่อมโยงหรือติดต่อกับฐานข้อมูล
userobject	เป็น Object ที่ใช้ในการติดต่อกับผู้ใช้งาน โดยเป็น window เสมือน
structure	เป็นโครงสร้างของข้อมูลที่ใช้ในการประกาศค่าตัวแปรต่างๆที่ใช้ในระบบ
function	เป็น Object ที่ใช้ในการสร้างฟังก์ชันต่างๆ เพื่อให้หน้าจอหรือ window ต่างๆเรียกใช้งานได้

การพัฒนาโดยโปรแกรม Sybase Power Builder 12.0 จะเริ่มจากการสร้าง workspace หรือพื้นที่ทำงานก่อน จากนั้นจึงทำการสร้าง target โดยขณะที่เราสร้าง target นั้น โปรแกรมจะทำการสร้าง application ให้โดยอัตโนมัติ จากนั้นจึงทำการสร้าง library เพื่อรองรับการสร้าง Power Builder Object ไม่ว่าจะเป็น window datawindow function และอื่นๆ หลังจากทำการพัฒนาโปรแกรมเรียบร้อยแล้ว เมื่อทำการ Compile โปรแกรม File ที่เป็น library จะถูกสร้างให้เป็น Object ที่เป็นส่วนที่สำคัญในการใช้งาน โปรแกรมต่อไป โดยโปรแกรมจะสร้าง file ที่มีส่วนขยายเป็น .pbl ให้โดยอัตโนมัติ ทั้งนี้เพื่อให้ง่ายต่อการปรับปรุงแก้ไขโปรแกรมในอนาคตต่อไปนั่นเอง

ในระบบสารสนเทศเพื่อการจัดการร้าน ส.อะไหล่ จังหวัดเพชรบูรณ์ จะมีการสร้าง library ที่ใช้ในการพัฒนาระบบดังนี้

ตาราง 5.2 แสดง library ของระบบสารสนเทศเพื่อการจัดการร้าน ส.อะไหล่ จังหวัดเพชรบูรณ์

Library	คำอธิบาย
app_alai.pbl	เป็น library ที่ใช้เก็บ Application และ userobject ที่ติดต่อกับฐานข้อมูล
comsrv.pbl	เป็น library ที่ใช้ในการเก็บ Object หน้าจอเข้าระบบ หน้าจออนุญาตการทำรายการ และ userobject ที่ใช้ในระบบ

ตาราง 5.2 แสดง library ของระบบสารสนเทศเพื่อการจัดการร้าน ส.อะไหล่ จังหวัดเพชรบูรณ์
(ต่อ)

Library	คำอธิบาย
constant.pbl	เป็น library ที่ใช้ในการเก็บ Object เกี่ยวกับการจัดการหน้าจอบระบบและ ผู้ใช้งานระบบ
datawindows prinslip.pbl	เป็น library ที่ใช้ในการเก็บ datawindow ที่ใช้ในการออกใบทำการ ต่างๆ
datawindows table.pbl	เป็น library ที่ใช้ในการเก็บ datawindow ตารางของฐานข้อมูลเพื่อใช้ใน การอัปเดตข้อมูลลงฐานข้อมูล
detail.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับรายละเอียดสินค้า และคู่ค้า
dropdowndata windows.pbl	เป็น library ที่ใช้ในการเก็บ datawindow เพื่อแสดง drop down datawindow ของตารางที่ต้องการโชว์รายการบางค่า
function.pbl	เป็น library ที่ใช้ในการเก็บ function ต่างๆ ของระบบ
processororder.pbl	เป็น library ที่ใช้ในการเก็บ object การประมวลผลจุดสั่งซื้อและยอด สั่งซื้อ
reference.pbl	เป็น library ที่ใช้ในการเก็บ object ข้อมูลพื้นฐานของระบบ
search.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับการค้นหาข้อมูลของระบบ
slipbuy.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับใบทำการสั่งซื้อและใบ ทำการการซื้อสินค้าของระบบ
slipcancel.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับการยกเลิกใบทำการการ
sliploanpay.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับใบทำการชำระหนี้
slipreceive.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับใบทำการการรับคืนสินค้า
slipsale.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับใบทำการการขาย
slipsent.pbl	เป็น library ที่ใช้ในการเก็บ object ที่เกี่ยวกับใบทำการการส่งคืน
scommon.pbl	เป็น library ที่ใช้ในการเก็บ object ที่สร้างขึ้นเพื่อเป็นหน้าจอหลักของ ระบบ และหน้าจอพื้นฐานของ object ต่างๆ ในระบบ เพื่อใช้คุณสมบัติ การถ่ายทอด

5.3.2 การหาจุดสั่งซื้อ และจำนวนสั่งซื้อ สามารถแสดงคำสั่งที่ใช้ในการเขียนโปรแกรมได้ ดังนี้

```

string    ls_productno
string    ls_sproductno , ls_eproductno , ls_sproducttypecode , ls_eproducttypecode
integer   li_stepvalue , li_daycal , li_orderpointamt
long      ll_index , ll_count
dec{7}    ldc_balanceqty , ldc_eoqty , ldc_balancereserve , ldc_opqty
dec{7}    ldc_ordercostamt , ldc_inventoryvalue , ldc_costavgamt
dec{7}    ldc_sumbalanceqty , ldc_sumcostamt
datetime  ldtm_sslipdate , ldtm_eslipdate

// จำนวนวันที่สั่งแล้วจะได้ของ ( ระยะเวลาในการสั่งซื้อ )
li_orderpointamt    = integer( ProfileString ( "C:\salai\salai01std\market.ini" , "OP" , "orderpoint_amt" , "7" ) )
// ค่าใช้จ่ายในการสั่งซื้อต่อครั้ง
ldc_ordercostamt    = dec( ProfileString ( "C:\salai\salai01std\market.ini" , "EOQ" , "ordercost_amt" , "10" ) )
// ค่าเก็บรักษา ที่ % ของราคาทุน
ldc_inventoryvalue  = dec( ProfileString ( "C:\salai\salai01std\market.ini" , "EOQ" , "inventory_value" , "10" ) )

// ค่าเก็บเป็นจำนวนอาหาร 100 เพื่อคิดเป็นร้อยละ
ldc_inventoryvalue  /= 100

ls_sproductno      = dw_master.object.sproduct_no[ 1 ]
ls_eproductno      = dw_master.object.eproduct_no[ 1 ]
ls_sproducttypecode = dw_master.object.sproducttype_code[ 1 ]
ls_eproducttypecode = dw_master.object.eproducttype_code[ 1 ]
ldtm_sslipdate     = dw_master.object.sslip_date[ 1 ]
ldtm_eslipdate     = dw_master.object.eslip_date[ 1 ]

// นับจำนวนตามวันที่เลือกจากจอร์บบ
li_daycal          = daysafter( date(ldtm_sslipdate) , date(ldtm_eslipdate) )
li_daycal++

if dw_detail.retrieve( gs_companyid , ls_sproducttypecode , ls_eproducttypecode , ls_sproductno , ls_eproductno ) < 0 then
    messagebox( 'แจ้งเตือน' , 'ไม่สามารถดึงรายการทะเบียนสินค้าได้กรุณาตรวจสอบ' )
    return failure
end if

ll_count          = dw_detail.rowcount()

```



```

li_stepvalue          = of_setprogressbar( ll_count, 1 )

// คำนวณ EOQ
for ll_index = 1 to ll_count

    if mod ( ll_index , li_stepvalue ) = 0 then hpb_proc.stepit()

    dw_detail.selectrow( ll_index, true )
    dw_detail.scrolltorow( ll_index )

    // รหัสสินค้า
    ls_productno      = dw_detail.object.product_no[ ll_index ]

    // -----
    // จำนวนที่ใช้ต่อเดือน
    // ldc_sumbalanceqty คือ ยอดขายสินค้าออกจากคลังโดยดึงข้อมูลจากรายการเคลื่อนไหว
    // ldc_sumcostamt คือ ราคาทุนรวมต่อหน่วย ณ รายการเคลื่อนไหวนั้นๆ

    select      sum( item_qty ) ,          sum( cost_amt )
    into        :ldc_sumbalanceqty ,      :ldc_sumcostamt
    from        stockcard
    where       company_id                = :gs_companyid
    and         sliptype_code              in (      select  a.sliptype_code
                                                    from      sliptype a
                                                    where     a.company_id  = stockcard.company_id
                                                    and       a.sliptype_sign = -1 )

    and         product_no                 = :ls_productno
    and         slip_date                   between :ldtm_sslipdate and :ldtm_eslipdate
    and         stockcard_status           = 1;

    if isnull( ldc_sumbalanceqty ) or ldc_sumbalanceqty < 0 then ldc_sumbalanceqty = 0
    if isnull( ldc_sumcostamt ) or ldc_sumcostamt < 0 then ldc_sumcostamt = 0

    // จำนวนใช้ต่อเดือน = ( จำนวนทั้งหมดที่ขาย / ช่วงจำนวนวันที่เลือก ) * 30 วัน
    ldc_balanceqty      = ( ldc_sumbalanceqty / li_daycal ) * 30
    // จบจำนวนที่ใช้ต่อเดือน
    // -----

```

```

// -----
// ต้นทุนวัสดุเฉลี่ย
if ldc_sumbalanceqty <= 0 then
// ถ้าไม่มีจำนวนขายทั้งหมด ต้องกำหนดทุนเฉลี่ยเป็น 1 เพราะต้นทุนเฉลี่ยเป็นตัวหารของ EOQ จะทำให้ error
    ldc_costavgamt = 1
else
// ต้นทุนเฉลี่ย = ราคาทุนทั้งหมดที่ขาย / จำนวนทั้งหมดที่ขาย
    ldc_costavgamt = ldc_sumcostamt / ldc_sumbalanceqty
end if
// จบต้นทุนวัสดุเฉลี่ย
// -----
// การคำนวณ EOQ = รุท ( 2 * DS ) / ( IC )
// D = จำนวนที่ใช้ต่อเดือน : ldc_balanceqty
// S = ค่าใช้จ่ายในการจัดซื้อต่อครั้ง (บาท) : ldc_ordercostamt
// I = ค่าใช้จ่ายในการเก็บรักษาคิดเป็นกึ่งเปอร์เซ็นต์ของราคาทุน : ldc_inventoryvalue
// C = ราคาต้นทุน(บาท) ในการคำนวณจะใช้ราคาต้นทุนเฉลี่ยจากช่วงที่ทำการประมวลผล : ldc_costavgamt
ldc_eoqty = (( 2 * ldc_balanceqty ) * ldc_ordercostamt ) / ( ldc_costavgamt * ldc_inventoryvalue )
// ถอดรุท
ldc_eoqty = sqrt ( ldc_eoqty )
// ปัดเป็นจำนวนเต็ม
ldc_eoqty = round( ldc_eoqty , 0 )

dw_detail.object.calorder_amt[ ll_index ] = ldc_eoqty

st_proc_num.text = "คำนวณ EOQ : " + string ( ll_index , '###0' ) + " / " + string ( ll_count , '###0' )

dw_detail.selectrow( ll_index , false )

next
// จบการคำนวณ EOQ
// คำนวณ จุดสั่งซื้อ
for ll_index = 1 to ll_count

if mod ( ll_index , li_stepvalue ) = 0 then hpb_proc.stepit()

dw_detail.selectrow( ll_index , true )

dw_detail.scrolltorow( ll_index )

```

```

// รหัสสินค้า
ls_productno      = dw_detail.object.product_no[ ll_index ]
// จำนวนสินค้าสำรองเมื่อรอการสั่งซื้อสินค้า
ldc_balancereserve = dw_detail.object.balance_reserve[ ll_index ]

if isnull( ldc_balancereserve ) then ldc_balancereserve = 0

// -----
// จำนวนที่ใช้ต่อวัน
// ldc_sumbalanceqty คือ ยอดจ่ายสินค้าออกจากคลัง โดยดึงข้อมูลจากรายการเคลื่อนไหว
// ldc_sumcostamt คือ ราคาทุนรวมต่อหน่วย ณ รายการเคลื่อนไหวนั้นๆ
select      sum( item_qty ) ,          sum( cost_amt )
into        :ldc_sumbalanceqty ,      :ldc_sumcostamt
from        stockcard
where       company_id                = :gs_companyid
and         sliptype_code              in (
select      a.sliptype_code
from        sliptype a
where       a.company_id              = stockcard.company_id
and         a.sliptype_sign           = -1 )
and         product_no                = :ls_productno
and         slip_date                 between :ldtm_sslipdate and :ldtm_eslipdate
and         stockcard_status          = 1;

if isnull( ldc_sumbalanceqty ) or ldc_sumbalanceqty < 0 then ldc_sumbalanceqty = 0
if isnull( ldc_sumcostamt ) or ldc_sumcostamt < 0 then ldc_sumcostamt = 0

// จำนวนใช้ต่อวัน      = ( จำนวนทั้งหมดที่ขาย / จำนวนวันที่เลือก )
ldc_balanceqty        = ( ldc_sumbalanceqty / li_daycal )
// จบจำนวนที่ใช้ต่อวัน
// -----

// คำนวณจุดสั่งซื้อ
// R = ( L * V ) + S
// R = จุดสั่งซื้อ          : ldc_opqty
// L = ระยะเวลาในการสั่งซื้อ(วัน)      : li_orderpointamt
// V = อัตราการใช้สินค้าต่อวัน        : ldc_balanceqty
// S = สินค้าคงเหลือสำรองหรือสินค้าคงเหลือเพื่อความปลอดภัย          : ldc_balancereserve
ldc_opqty              = ( ldc_balanceqty * li_orderpointamt ) + ldc_balancereserve
// ปัดให้เป็นจำนวนเต็ม
ldc_opqty              = round( ldc_opqty , 0 )

```

```


st_proc_num.text      = "จำนวน จุดตั้งชื่อ : " + string (ll_index , '###0') + " / " + string (ll_count , '###0')

dw_detail.object.calminpoint_qty[ ll_index ] = ldc_opqty

dw_detail.selectrow( ll_index, false )

next
// จบการหาจุดตั้งชื่อ

```



ลิขสิทธิ์มหาวิทยาลัยเชียงใหม่
 Copyright© by Chiang Mai University
 All rights reserved